Question:

A factory producing energy drinks has branches in five states. The production capacity of each branch of the factory (in tons) in the states of California, New York, Washington, Texas, and Florida is as follows.

	CA	NY	WA	TX	FL
Capacity	300	200	300	200	400

Before packaging, the drinks are stored in one of three warehouse locations. The cost of producing each ton of the energy drink and delivering it to one of the three warehouse locations is presented in the table in terms of hundred dollars.

	Warehouse1	Warehouse2	Warehouse3
CA	8	10	12
NY	7	5	7
WA	8	6	7
TX	5	6	7
FL	7	6	5

Four major customers, as the primary and fixed consumers of this brand of energy drink, incur a delivery cost per ton as specified in the table.

	Customer1	Customer2	Customer3	Customer4
Warehouse1	40	80	90	50
Warehouse2	70	70	60	80
Warehouse3	80	30	50	60

Additionally, the demand for each of these customers is as follows.

	Customer1	Customer2	Customer3	Customer4
Demand	200	300	150	250

A) Define an appropriate transportation model that allows this energy drink brand to minimize its costs.

B) Now, assuming the numbers in Table 4 represent the annual demand for this brand and the fixed establishment costs for each branch and warehouse are provided in thousands of dollars, use an integer programming approach to minimize the total cost of this energy drink brand. Implement and run your model using the GAMS software. Analyze the results obtained from the software output and provide recommendations for the brand owner.

Solution (By Sajjad Abed):

A:

Mathematical Model:

$$Min\ Z = 100 * \sum_{i=1}^{5}\sum_{j=1}^{3} P_{i,j} * PC_{i,j} + \sum_{j=1}^{3}\sum_{k=1}^{4} S_{j,k} * SC_{j,k}$$

$s.t.$

$$\sum_{j=1}^{3} P_{i,j} \leq C_i \qquad for\ i\ in(1,2,3,4,5)$$

$$\sum_{j=1}^{3} S_{j,k} = D_k \qquad for\ k\ in(1,2,3,4)$$

$$\sum_{k=1}^{4} S_{j,k} \leq \sum_{i=1}^{5} P_{i,j} \qquad for\ j\ in\ (1,2,3)$$

$$S_{j,k} \geq 0 \qquad for\ j\ in\ (1,2,3) \text{ and k in } (1,2,3,4)$$

$$P_{i,j} \geq 0 \qquad for\ i\ in\ (1,2,3,4,5) \text{ and j in } (1,2,3)$$

Variables and Parameters:

Quantity of the beverage produced in city i and sent to warehouse j (Variable)	$P_{i,j}$
Cost of producing one ton of beverage at factory i and delivering it to warehouse j (Parameter):	$PC_{i,j}$
Quantity of customer demand fulfilled by warehouse j for customer k (Variable):	$S_{j,k}$
Cost of delivering one ton of beverage from warehouse j to customer k (Parameter):	$SC_{j,k}$
Capacity of factory i (Parameter):	C_i
Quantity of request from customer k (Parameter):	D_k

B:

$$Min\ Z = 100 * \sum_{i=1}^{5}\sum_{j=1}^{3} P_{i,j} * PC_{i,j} + \sum_{j=1}^{3}\sum_{k=1}^{4} S_{j,k} * SC_{j,k} + 1000 * \sum_{i=1}^{5} y_i * yC_i + 1000 * \sum_{j=1}^{3} w_j * wC_j$$

$s.t.$

$$\sum_{j=1}^{3} P_{i,j} \leq\ y_i * C_i \qquad for\ i\ in (1,2,3,4,5)$$

$$\sum_{j=1}^{3} S_{j,k} =\ D_k \qquad for\ k\ in (1,2,3,4)$$

$$\sum_{k=1}^{4} S_{j,k} \leq \sum_{i=1}^{5} P_{i,j} \qquad for\ j\ in\ (1,2,3)$$

$$\sum_{i=1}^{5} P_{i,j} \leq\ w_i * 900 \qquad for\ i\ in\ (1,2,3,4,5)$$

$w_i \in \{0,1\}$

$y_i \in \{0,1\}$

$S_{j,k} \geq 0$ $\qquad for\ j\ in$ (1,2,3) and k in (1,2,3,4)

$P_{i,j} \geq 0$ $\qquad for\ i\ in$ (1,2,3,4,5) and j in (1,2,3)

In addition to the six items introduced in the previous section, we also have the following variables and parameters in this section:

Activation of factory i (Binary variable: 0 or 1): yi

Cost of activating factory i (Parameter): yCi

Activation of warehouse j (Binary variable: 0 or 1): wi

Cost of activating warehouse j (Parameter): wCi

GAMS code:

```
Sets
i Companies /CA, NY, WA,TX, FL/
j Warehouses/Warehouse1,Warehouse2,Warehouse3/
k Customers /Customer1,Customer2,Customer3,Customer4/;

parameters C(i) Capacity of each company;
$call gdxxrw.exe C:\Users\sabed\Documents\gamsdir\projdir\Phase2_Data_OR2.xlsx par=C Rng=Capacity!C5:G6 cdim=1
$gdxin Phase2_Data_OR2.gdx
$load C
$gdxin
;
parameters D(k) Demand of each customer;
$call gdxxrw.exe C:\Users\sabed\Documents\gamsdir\projdir\Phase2_Data_OR2.xlsx par=D Rng=Demand!C7:F8 cdim=1
$gdxin Phase2_Data_OR2.gdx
$load D
$gdxin
;
parameters PC(i,j) Cost of Production and inventory;
$call gdxxrw.exe C:\Users\sabed\Documents\gamsdir\projdir\Phase2_Data_OR2.xlsx par=PC Rng=Cost_W!B4:E9 rdim=1 cdim=1
$gdxin Phase2_Data_OR2.gdx
$load PC
$gdxin
;
parameters SC(j,k) Cost of Production and inventory;
$call gdxxrw.exe C:\Users\sabed\Documents\gamsdir\projdir\Phase2_Data_OR2.xlsx par=SC Rng=Cost_C!B5:F8 rdim=1 cdim=1
$gdxin Phase2_Data_OR2.gdx
$load SC
$gdxin
;
parameters yC(i) fixed cost of each company;
$call gdxxrw.exe C:\Users\sabed\Documents\gamsdir\projdir\Phase2_Data_OR2.xlsx par=yC Rng=Cost_F!B5:C9 rdim=1
$gdxin Phase2_Data_OR2.gdx
$load yC
$gdxin
;
parameters wC(j) fixed cost of each warehouse;
$call gdxxrw.exe C:\Users\sabed\Documents\gamsdir\projdir\Phase2_Data_OR2.xlsx par=wC Rng=Cost_F!B10:C12 rdim=1
$gdxin Phase2_Data_OR2.gdx
$load wC
$gdxin
;
```

```
Variables
z Goal
y(i) company initialize
w(j) warehouse initialize
P(i,j) tonnage of production
S(j,k) tonnage of Supply
;
positive variable
P
S;
binary variable
y
w;

equations
cost objective func
capacity_limit(i)
demand_supplies(k)
inventory(j)
warehouse_initialize(j);
cost .. z=e=100*sum((i,j),P(i,j)*PC(i,j))+sum((j,k),S(j,k)*SC(j,k))+1000*sum(i,y(i)*yC(i))+1000*sum(j,w(j)*wC(j));
capacity_limit(i) .. sum(j,P(i,j))=l=C(i)*y(i);
demand_supplies(k) .. sum(j,S(j,k))=e=D(k);
inventory(j) .. sum(k,S(j,k))=l=sum(i,P(i,j));
warehouse_initialize(j) .. sum(i,P(i,j))=l=w(j)*900;

model main/all/;
solve main using mip minimizing z;
display Z.l,P.l,S.l,y.l,w.l,P.m,S.m,z.m;

execute_unload"Phase2_Data_OR2.gdx" z, P, S;
execute'gdxxrw.exe  Phase2_Data_OR2.gdx o=Phase2_Data_OR2.xlsx var=z Rng=z!b2'
execute'gdxxrw.exe  Phase2_Data_OR2.gdx o=Phase2_Data_OR2.xlsx var=P Rng=p!b2'
execute'gdxxrw.exe  Phase2_Data_OR2.gdx o=Phase2_Data_OR2.xlsx var=S Rng=s!b2'
```

Result:

```
----      71 VARIABLE z.L                    =   725000.000  Goal

----      71 VARIABLE P.L  tonnage of production

    Warehouse2  Warehouse3

NY     200.000
WA     300.000
FL                 400.000


----      71 VARIABLE S.L  tonnage of Supply

             Customer1   Customer2   Customer3   Customer4

Warehouse2     200.000                 150.000     150.000
Warehouse3                 300.000                 100.000


----      71 VARIABLE y.L  company initialize

NY 1.000,    WA 1.000,    FL 1.000


----      71 VARIABLE w.L  warehouse initialize

Warehouse2 1.000,    Warehouse3 1.000


----      71 VARIABLE P.M  tonnage of production

    Warehouse1  Warehouse2  Warehouse3

CA     170.000     400.000     580.000
NY     170.000                 180.000
WA     170.000                  80.000
TX                 130.000     210.000
FL     190.000     120.000

----      71 VARIABLE S.M  tonnage of Supply

             Customer1   Customer2   Customer3   Customer4

Warehouse1                  60.000      60.000        EPS
Warehouse2                  20.000
Warehouse3      30.000                  10.000


----      71 VARIABLE z.M                    =        0.000  Goal
```

Z

total	725000

S

	Customer1	Customer2	Customer3	Customer4
Warehouse2	200		150	150
Warehouse3		300		100

P

	Warehouse2	Warehouse3
NY	200	
WA	300	
FL		400